

- [NAME](#)
- [AUTHOR](#)
- [SYNOPSIS](#)
- [DESCRIPTION](#)
 - [SEARCH SCHEMA](#)
 - [LOADING](#)
 - [SEARCHING](#)
 - [PROBLEMS](#)
 - [SUMMARY](#)

NAME

A (Modest) Proposal for Replacing Gramene's Quick Search

AUTHOR

Ken Youens-Clark <kclark@cshl.edu>

SYNOPSIS

I offer an idea for a comprehensive search when most of your data is locked away in databases.

<http://dev.gramene.org/db/searches/unisearch>

DESCRIPTION

Gramene's current "quick" search isn't all that quick:

http://www.gramene.org/db/searches/browser?search_type=All&query=c86&RGN=on

This script takes the "query" and uses Perl and LWP to query each module's search scripts with the added argument of "table_only=1" which instructs the scripts to leave out the header and footer wrappers, returning only the "table" of results. It's not entirely quick, but it does the job and maintains a level of consistency at least in that any search result for a module here will be the same as if the user had gone to the specific module search interface.

What I think could be improved:

- **Allow more tables and fields to be searched**
- **Use a search engine for more faster searches**
- **Use a search engine for more interesting matches (stems, soundex)**
- **Present the results in a uniform, Googlish way**

While developing Gramene's diversity module around the GDPDM schema:

<http://www.maizegenetics.net/gdpdm/>

I realized that a field-by-field, table-by-table search in a schema of 30+ tables would be inefficient (to say the least), so

I had an idea to unfield the data and cram everything that needed to be searched into a single MySQL table that would use the FULLTEXT index:

<http://dev.mysql.com/doc/refman/5.0/en/fulltext-search.html>

I decided to port the idea over for Gramene and added the facility to also index all the static HTML on our site. I introduced the idea to the ``gramene-dev" list many months ago; however, the implementaion left some to be desired, and the idea was shelved. Recently I decided to dust it off, having learned some things that I wanted to apply.

SEARCH SCHEMA

First, let me show you the schema:

```
CREATE TABLE 'doc_search' (  
  'path' varchar(255) NOT NULL default '',  
  'title' varchar(255) default NULL,  
  'contents' text,  
  UNIQUE KEY ('path'),  
  FULLTEXT KEY ('path','title','contents')  
) ENGINE=MyISAM;  
  
CREATE TABLE 'module_search' (  
  'module_name' varchar(40) NOT NULL default '',  
  'table_name' varchar(40) NOT NULL default '',  
  'record_id' int(10) unsigned NOT NULL default '0',  
  'record_text' text,  
  UNIQUE KEY ('module_name','table_name','record_id'),  
  KEY ('module_name','table_name'),  
  FULLTEXT KEY ('record_text')  
) ENGINE=MyISAM;
```

LOADING

The ``doc_search" is pretty simple. I crawl the Gramene docroot and find all the documents I want to index, including HTML, Word Documents, and PDFs. (PDFs are actually pretty horrendous to index as the text comes out pretty mangled, e.g., the word ``the" might come out like ``t he" if the text was, say, justified. Luckily, it seems Gramene's PDFs are alternate formats of other texts like HTML, which are just fine.) For each document, I strip out any ugly stuff like HTML tags and put the contents into the ``contents" field (duh). There is a FULLTEXT index on the title, path, and contents of the document.

The module search is just a little more interesting, and I'll have to give you some background real quick-like: Gramene has a number of ``modules" -- discrete units of databases and code that represent some type of data we store like ``markers" or ``QTLs". A ``module" is defined in the ``<modules>" section of ``gramene.conf" and has a corresponding ``<section>" where it defines how to connect to its database. This is important for the ``Gramene::DB" module to work which is used by the Gramene::CDBI modules. A while back, the Gramene team decided to start using a standardized object-relational model (ORM). We settled on the CPAN module ``Class::DBI":

<http://search.cpan.org/dist/Class-DBI/>

Our namespace is called ``Gramene::CDBI", and to use, say the CDBI interface to the markers db, you just say ``use Gramene::CDBI::Markers."

OK, so back to loading the modules search. I run through each table in the schema for each module and extract all the ``text" of each record. (The text is basically anything in a field that isn't the primary key or a foreign key, and I know which are which because of the details that are in the Gramene::CDBI classes, which were generated directly and automatically from the MySQL schemas [see ``bin/mk-gcdbi.pl" and ``templates/class-dbi/gramene-cdbi.tmpl" under

```
``/usr/local/gramene/"].)
```

Curators can limit which tables in a schema are indexed in the ``gramene.conf" section for ``<gramene_search>" using the ``<limit_index>". E.g., here's the line for the qtl db:

```
<limit_index>
  qtl  index_only=qtl+qtl_trait,qtl_trait_synonym
</limit_index>
```

This says to index only the ``qtl" table and to aggregate (the ``+" sign) the tables ``qtl_trait" and ``qtl_trait_synonym" for each ``qtl" record. (Use a semi-colon to separate the tables+aggregates.) This ``aggregation" of data is one of the ideas I got after working more with the GDPDM/diversity schema -- sometimes your data is very normalized and the thing being sought is actually pretty far away from the concept that the user cares to see. So in the above example, if I found a search string in the ``qtl_trait_synonym" table, I would return the find in the context of which QTLs matched, not which trait synonyms. FWIW, it seems to me that it's best to index only on table per module and to aggregate as much data as possible around that.

So now, this should make the the script that loads the schema understandable:

```
1: #!/usr/local/bin/perl
2:
3: # vim: tw=78: sw=4: ts=4: et:
4:
5: # $Id: load-gramene-search.pl,v 1.2 2007/02/01 19:46:43 kclark Exp $
6:
7: use strict;
8: use warnings;
9: use CAM::PDF;
10: use DBI;
11: use Data::Dumper;
12: use English qw( -no_match_vars );
13: use File::Basename;
14: use File::Extract::PDF;
15: use File::Extract::Result;
16: use File::Find::Rule;
17: use File::Spec::Functions;
18: use Getopt::Long;
19: use Gramene::Config;
20: use Gramene::DB;
21: use Gramene::Utils qw( commify get_logger table_name_to_gramene_cdbi_class );
22: use HTML::HeadParser;
23: use HTML::Strip;
24: use IO::Prompt;
25: use PDF::API2;
26: use Perl6::Slurp;
27: use Pod::Usage;
28: use Readonly;
29:
30: $| = 1;
31:
32:
33: delete @ENV{qw(IFS CDPATH ENV BASH_ENV)};
34: $ENV{'PATH'} = '/bin:/usr/bin:/usr/local/bin';
35:
36: Readonly my $ANTIWORD => '/usr/local/bin/antiword';
37: Readonly my $COMMA      => q{, };
38: Readonly my $DBL_COLON => q{::};
```

```

39: Readonly my $EMPTY_STR => q{};
40: Readonly my $NL        => qq{\n};
41: Readonly my $SLASH     => q{/};
42: Readonly my $SPACE     => q{ };
43: Readonly my $VERSION   => sprintf '%d.%02d',
44:                               qq$Revision: 1.2 $ =~ /(\d+)\.(\d+)/;
45:
46: my $modules            = $EMPTY_STR;
47: my $list_modules = 0;
48: my ( $help, $man_page, $show_version );
49: GetOptions(
50:     'l|list'      => \$list_modules,
51:     'm|modules:s' => \$modules,
52:     'help'        => \$help,
53:     'man'         => \$man_page,
54:     'version'     => \$show_version,
55: ) or pod2usage(2);
56:
57: if ( $help || $man_page ) {
58:     pod2usage({
59:         -exitval => 0,
60:         -verbose => $man_page ? 2 : 1
61:     });
62: };
63:
64: if ( $show_version ) {
65:     my $prog = basename( $PROGRAM_NAME );
66:     print "$prog v$VERSION\n";
67:     exit 0;
68: }
69:
70: my %process      = map { s/^\s+|\s+//g; lc $_, 1 } split /\s/, $modules;
71: my $config       = Gramene::Config->new;
72: my $modules_conf = $config->get('modules');
73: my $search_conf  = $config->get('gramene_search');
74: ( my $db_name    = $search_conf->{'db_dsn'} ) =~ s/.*://;
75: my @modules      = ref $modules_conf->{'module'} eq 'ARRAY'
76:     ? @{$modules_conf->{'module'}}
77:     : ( $modules_conf->{'module'} );
78: my %valid_module = map { $_, 1 } ( @modules, 'documents' );
79:
80: if ( $list_modules ) {
81:     print join $NL,
82:         'Valid modules:',
83:         ( map { " $_" } sort keys %valid_module ),
84:         $EMPTY_STR;
85:     exit 0;
86: }
87:
88: if ( %process ) {
89:     my @bad = grep { !exists $valid_module{ $_ } } keys %process;
90:
91:     if ( @bad ) {
92:         my $bad  = join $COMMA, sort @bad;
93:         my $valid = join $COMMA, sort @modules;
94:         die "Bad modules ($bad).\nPlease choose from:\n$valid\n";
95:     }
96:

```

```

97:     my $ok = prompt -yn,
98:         sprintf( "OK to recache %s in '$db_name'? ",
99:             %process
100:             ? join($COMMA, sort keys %process)
101:             : join($COMMA, sort @modules)
102:         )
103:     ;
104:
105:     if ( !$ok ) {
106:         print "Exiting.\n";
107:         exit 0;
108:     }
109: }
110: else {
111:     %process = map { $_, 1 }
112:         prompt -menu => [ (sort keys %valid_module), 'ALL' ], 'Which module?';
113:
114:     if ( $process{'ALL'} ) {
115:         %process = %valid_module;
116:     }
117: }
118:
119: my $logger = get_logger();
120: $logger->info("Reloading search db '$db_name'");
121:
122: my $db = Gramene::DB->new('gramene_search');
123:
124: my $insert_sql = q[
125:     insert
126:     into    module_search (module_name, table_name, record_id, record_text)
127:     values (?, ?, ?, ?)
128: ];
129:
130: my $num_modules = 0;
131: my $num_records = 0;
132:
133: for my $module ( @modules ) {
134:     if ( %process ) {
135:         next if !defined $process{ lc $module };
136:     }
137:
138:     my ( %index_only, %skip );
139:     if ( my $search_limit = $search_conf->{'limit_index'}{ $module } ) {
140:         my ( $directive, $tables ) = split /=/, $search_limit;
141:         if ( $directive !~ /^(index_only|skip)$/ ) {
142:             die "Bad directive ($directive) in limit clause ($search_limit).\n";
143:         }
144:
145:         for my $table ( split /\s/, $tables ) {
146:             if ( $directive eq 'skip' ) {
147:                 $skip{ $table } = 1;
148:             }
149:             else {
150:                 if ( $table =~ /\s+(\w+)\s+(.+)\s/ ) {
151:                     $index_only{ $1 } = [ split /\s/, $2 ];
152:                 }
153:                 else {
154:                     $index_only{ $table } = 1;

```

```

155:         }
156:     }
157: }
158: }
159:
160: my $CDBI = join $DBL_COLON, 'Gramene', 'CDBI', ucfirst(lc($module));
161: my $CDBIPM = join $SLASH, 'Gramene', 'CDBI', ucfirst(lc($module)) . '.pm';
162:
163: eval { require $CDBIPM };
164:
165: if ( my $err = $@ ) {
166:     $logger->debug("Error requiring $CDBIPM: $err");
167:     die $err;
168: }
169:
170: $num_modules++;
171:
172: print "Removing previous data for module '$module'.\n";
173: $db->do(
174:     'delete from module_search where module_name=?', {},
175:     ( $module )
176: );
177:
178: TABLE:
179: for my $table ( $CDBI->represented_tables ) {
180:     if (
181:         $skip{ $table }
182:         || ( %index_only && !exists $index_only{ $table } )
183:     ) {
184:         next TABLE;
185:     }
186:
187:     my $class = table_name_to_gramene_cdbi_class( $module, $table );
188:     my $id_field = $class->columns('Primary');
189:
190:     if ( !$id_field ) {
191:         print STDERR "No PK in $class\n";
192:         next TABLE;
193:     };
194:
195:     my @has_a = keys %{ $class->meta_info('has_a') || {} };
196:     my $skip = join '|', $id_field, @has_a;
197:     my @columns = grep { !/($skip)/ } $class->columns('All');
198:
199:     if ( scalar @columns == 0 ) {
200:         print STDERR "No usable columns in $class.\n";
201:         next TABLE;
202:     };
203:
204:     my $count
205:         = $class->db_Main->selectrow_array("select count(*) from $table");
206:
207:     printf "\rProcessing %s record%s in '%s.%s'\n",
208:         commify($count),
209:         $count == 1 ? '' : 's',
210:         $module,
211:         $table;
212:

```

```

213: my $sth = $class->db_Main->prepare("select $id_field from $table");
214: $sth->execute;
215:
216: my @index_also
217:     = $index_only{ $table } && ref $index_only{ $table } eq 'ARRAY'
218:     ? @{ $index_only{ $table } }
219:     : ();
220:
221: my ( $c, $i ) = ( 1, 1 );
222: my $max = 50;
223: while ( my $id = $sth->fetchrow_array ) {
224:     my $rec = $class->retrieve( $id );
225:     if ( $i == $max ) {
226:         $i = 0;
227:     }
228:     else {
229:         print "$c ", ( '.' x int(100*($c/$count)/2) ), "\r";
230:         $i++;
231:     }
232:     $c++;
233:
234:     $num_records++;
235:     my $text
236:         = join $SPACE,
237:         map { defined $_ ? $_ : () }
238:         $rec->get(@columns);
239:
240:     OTHER_TABLE:
241:     for my $other_table ( @index_also ) {
242:         my $other_class
243:             = table_name_to_gramene_cdbi_class($module, $other_table);
244:         my $other_id_field = $other_class->columns('Primary');
245:
246:         if ( !$other_id_field ) {
247:             print STDERR "No PK in $class\n";
248:             next OTHER_TABLE;
249:         };
250:
251:         my @other_has_a
252:             = keys %{ $other_class->meta_info('has_a') || {} };
253:         my $other_skip
254:             = join '|', $other_id_field, @other_has_a;
255:         my @other_columns
256:             = grep { !/($other_skip)/ } $other_class->columns('All');
257:
258:         for my $other_obj ( $rec->get_related($other_table) ) {
259:             $text
260:                 .= join $SPACE,
261:                 $EMPTY_STR,
262:                 map { defined $_ ? $_ : () }
263:                 $other_obj->get(@other_columns);
264:         }
265:     }
266:
267:     next if !$text;
268:
269:     $text =~ s/^\s+|\s+$//g; # trim
270:     $text =~ s/\s+/ /g;      # collapse spaces

```

```

271:         $db->do( $insert_sql, {}, ( $module, $table, $rec->id, $text ) );
272:     }
273:     print "\n";
274: }
275: }
276:
277: my $num_docs = 0;
278: if ( $process{'documents'} ) {
279:     my $ok = prompt -yn, "OK to recache documents in '$db_name'? ";
280:
281:     if ( !$ok ) {
282:         print "Exiting.\n";
283:         exit 0;
284:     }
285:
286:     my $gr_conf = $config->get('gramene') or die "No 'gramene' config.\n";
287:     my $base_dir = $gr_conf->{'base_dir'} or die "No 'base_dir' setting.\n";
288:     my $doc_root = catdir( $base_dir, 'html' );
289:     my @files = File::Find::Rule->file()
290:         ->name('*.html', '*.pdf', '*.doc')
291:         ->in($doc_root);
292:
293:     print "Removing previous data for 'documents'.\n";
294:     $db->do('truncate table doc_search');
295:
296:     my $hs = HTML::Strip->new;
297:
298:     FILE:
299:     for my $file ( @files ) {
300:         ( my $path = $file ) =~ s/^\$doc_root//;
301:         print "Indexing $path\n";
302:
303:         my $contents = $EMPTY_STR;
304:         my $title = $EMPTY_STR;
305:
306:         if ( $file =~ /\.html$/ ) {
307:             my $text = slurp $file;
308:             $contents = $hs->parse( $text );
309:             $hs->eof;
310:
311:             my $p = HTML::HeadParser->new;
312:             $p->parse( $text );
313:             $title = $p->header('Title');
314:         }
315:         elsif ( $file =~ /\.doc$/ ) {
316:             $contents = ` $ANTIWORD $file `;
317:             my $xml = ` $ANTIWORD -x db $file `;
318:             if ( $xml =~ /<title>(.*?)</title>/ ) {
319:                 $title = $1;
320:             }
321:         }
322:         elsif ( $file =~ /\.pdf$/ ) {
323:             my $pdf = CAM::PDF->new( $file ) or do {
324:                 warn "No PDF: $CAM::PDF::errstr\n";
325:                 next FILE;
326:             };
327:
328:             for my $p ( 1..$pdf->numPages() ) {

```



```

329:         if ( my $page = $pdf->getPageText( $p ) ) {
330:             $contents .= $page;
331:         }
332:     }
333:
334:     my $pdf2 = PDF::API2->new( $file );
335:     my %info = $pdf2->info;
336:     $title   = $info{'Title'};
337: }
338: else {
339:     warn "Don't know what to do with '$file'\n";
340: }
341:
342: $contents =~ s/^\s+|\s+$//g;
343:
344: if ( !$title ) {
345:     my @lines = grep { /\w+/ } split /\n/, $contents;
346:     my $first_line_len = @lines ? length $lines[0] : 0;
347:     if ( $first_line_len > 1 && $first_line_len < 100 ) {
348:         $title = $lines[0];
349:     }
350:     else {
351:         $title = basename( $file );
352:         $title =~ s/\.*/ /;
353:         $title =~ s/_/ /g;
354:     }
355: }
356:
357: if ( $contents ) {
358:     $contents =~ s/^[[:ascii:]]//g;
359:     $contents =~ s/\s+/ /g;
360:
361:     $db->do(
362:         q[
363:             insert
364:             into   doc_search (path, title, contents)
365:             values (?, ?, ?)
366:         ],
367:         {},
368:         ( $path, $title, $contents )
369:     );
370:     $num_docs++;
371: }
372: }
373: }
374:
375: printf
376:     "Done, processed %s record%s in %s module%s, %s document%s.\n",
377:     commify($num_records),
378:     $num_records == 1 ? '' : 's',
379:     commify($num_modules),
380:     $num_modules == 1 ? '' : 's',
381:     commify($num_docs),
382:     $num_docs == 1 ? '' : 's',
383: ;

```

SEARCHING

So that's how all data from six Gramene modules and a couple thousand static documents data gets put into two search tables. It's really the more interesting part of the search, IMO, because it manipulates the schemas and data so abstractly. The CGI search isn't really all that interesting because it just takes the query string and use MySQL "MATCH AGAINST" for the FULLTEXT search to retrieve the record from each module.

I divide the search results by module and show the user a summary at the top of the number of hits for each with links to them. What I think you might find interesting is:

- You can control the fields displayed for each module via the "<list_columns>":

```
<list_columns>
  literature.reference=title,source.source_name,year
</list_columns>
```

Here I'm indicating that, when showing the results for the "literature" module's "reference" table, show just the "title" and "year" from that table plus the "source_name" from the record's "source" method (a CDBI method!). This is made possible by taking the "module_name" and "table_name" from the "module_search" results turning those into a Gramene::CDBI class name and then retrieving the object via the ORM, like so:

```
my $class = table_name_to_gramene_cdbi_class( $module, $object )
           or die "Invalid object ($module.$object)";
my $item  = $class->retrieve( $id ) or die "Bad id '$id'";
```

- You can indicate how to get to the module-specific search interfaces using the "<view_link>" template:

```
<view_link>
  genes.gene_gene="/db/genes/search_gene?id="
</view_link>
```

This says that the "View" link for a "gene_gene" record in the "genes" module should put the string in quotes before the "record_id" (the primary key value of the record. This, of course, means that the script needs to be able to work with PK values, and many of Gramene's modules primarily use accessions, so I had to hack a couple of scripts to recognize this additional parameter.

- Each record indicates the context of the search hits a la Google.

This last point took me a few hours to work out. (Those Google guys are really pretty smart, you know.) Anyway, here's the sub I wrote to return the portions matching the search strings:

```
sub match_context {
  my ( $find, $s ) = @_;

  for my $f ( @$find ) {
    $s =~ s/($f)/{{ $1 }}/gi;
  }

  my $pos      = -1;
  my $match_num = 0;
  my $max_matches = 7;
  my $last_match = 0;
  while ( ( $pos = index( $s, '}', $pos ) ) > -1 ) {
    $match_num++;
    $last_match = $pos;
  }
}
```

```

        if ( $match_num == 7 ) {
            $s = substr( $s, 0, $pos + 25 );
        }
        else {
            $pos++;
        }
    }

    if ( $match_num < $max_matches && length($s) - $last_match > 50 ) {
        $s = substr( $s, 0, $last_match + 25 );
    }

    $s =~ s/^.*(?:(?:\S+\s+){2}\S*[\s]{2})/...$1/xms;
    $s =~ s/
        ([\s]{2}\S*)
        ((?:\s+\S+){3})
        .{$MAX_CHARS_MATCH_CONTEXT,}?
        ((?:\s+\S+){3})
        (\S*[\s]{2})
        /$1$2...$3$4/gxms;

    $s =~ s/[ \s]{2}/<b>/g;
    $s =~ s/[ \s]{2}/<\/b>/g;

    return $s;
}

```

First I put double braces around each search hit, then I cut off the text after a maximum of seven matches. Then I remove any long bits of text between matches and try to add a couple words around them. Then I turn my double braces into HTML ``bold" tags.

This is all pretty neat, except that I'm not matching in quite the same way that MySQL's FULLTEXT engine does -- it can do nifty things like word stemming and soundex and similarity and misspellings, and I'm doing exact matches. However, it's better than nothing.

PROBLEMS

One drawback I'm dealing with is the enormity of Gramene's markers db, which currently has 13.5M markers with 38.4M marker synonyms. To load the markers db, I would have to aggregate millions of records (and a trial run did 150K in 12 hours) and add 13.5M records to the search schema. Given that most people search the marker names and that the markers schema was optimized around just such searches, it makes no sense to denormalize all this data and bloat the search db, thereby slowing it down. So I think I will probably just tap the markers search API directly when I get to including markers in the quick search.

Another possible problem is that this search might work differently from the module-specific searches, possibly confusing the user. This is especially true of the similarity-based matches that the FULLTEXT engine will find, e.g., a search for ``Texmont" will also return ``Rosemont," whereas a stricter field-based search common in the other interfaces would never find such a match.

SUMMARY

So I've tried to show you how I do a lot of fancy manipulation of data using very high-level interfaces that are generated by underlying code. Here's a little graph showing how all those early foundation pieces fit together:

```

mk-gcdbi.pl
|
v
SQL::Translator      gramene.conf

```

```
      |           ^  
      v           |  
MySQL db (markers)  Gramene::Config  
      |           ^  
      v           |  
Gramene::CDBI::Markers <-- Gramene::DB
```

Once those CDBI classes are in place, it makes it easy to abstract the logic for digesting the data into the search schema and then turning around and making the abstractions back into objects using the ORM for purposes of displaying them.

I hope you've enjoyed yourselves. Be sure to tip your waiter and bartender. I'll be here all week.